Line-by-Line Explanation of the Maple Code for the Key Vanishing Lemma (Compact Case)

May 9, 2026

1 Introduction

This Maple code proves a special case of the Key Vanishing Lemma for smooth surfaces in $\mathbb{P}^3$ (compact case). It verifies that for a generic surface $X \subset \mathbb{P}^3$ of degree $d = 17$ there are **no** nonzero invariant 2-jet differentials of weighted order $m = 6$ twisted by $\mathcal{O}_X(-6)$. The parameters are

$$m = 6, \qquad \mathsf{twist} = -6, \qquad d_{\text{aux}} = 16 \quad \text{(so the surface degree is } d_{\text{aux}} + 1 = 17\text{)}.$$

The algorithm enforces holomorphicity conditions that a global section must satisfy, first on the affine charts and then across the hyperplane at infinity. By solving large linear systems, all coefficients of a candidate differential are forced to zero. By semicontinuity, the vanishing extends to a generic surface of the same degree.

Flexibility of the parameters. The three essential numbers are set at the very top:

$$\texttt{m}, \qquad \texttt{twist}, \qquad \texttt{d}.$$

For a surface of degree D one takes $\texttt{d} = D - 1$. By adjusting `m`, `twist` and `d`, the same code verifies all cases required by Lemma 1.1 for surfaces of degree 17. The present run with $m = 6$, $\texttt{twist} = -6$, $d = 16$ is a concrete illustration.

Limitations.

- **Degree 16.** The same construction, with the appropriate choice of the auxiliary monomial in the surface equation, should in principle prove the key vanishing lemma for degree 16 as well. However, the resulting linear systems become too large for Maple to handle in a reasonable time. It is very likely that a translation of the algorithm into C^{++} combined with parallel computation would overcome this obstacle, but this is beyond the scope of the present Maple implementation.
- **Degree 15 and the case $(m, t) = (3, 1)$.** For a surface of degree 15 (i.e. $d = 14$), the same Fermat-type perturbation *cannot* prove the required vanishing. When running the code with these parameters, the algorithm finds a 1-dimensional space of non-zero sections for $(m, t) = (3, 1)$. To prove the key vanishing lemma for degree 15 one would need to start from a more involved surface equation (for example, containing several extra monomials). Such a modification is not covered by the present Maple code.

Now we proceed line by line through the Maple code.

2 Initialization

```
restart:
m := 6;
twist := -6;
d := 16;                degre := d + 1:
```

- `restart`: clears all previous definitions.
- `m := 6`: weighted order of the invariant 2-jet differential.
- `twist := -6`: the bundle is tensored with $\mathcal{O}_X(-6)$.
- `d := 16`: auxiliary integer; the true surface degree is `degre` $= d + 1 = 17$.

3 The surface equation

```
HUU := X^(d+1) + T^(d+1) + 1*X^floor((d+1)/2)*T^(d+1-floor((d+1)/2));
UU  := subs({T=1,X=x}, HUU);
```

- `HUU` defines the X-T part of the homogeneous equation $\widehat{R}(T,X,Y,Z) = X^{17}+T^{17}+X^8T^9+Y^{17}+Z^{17}$. Here `floor((d+1)/2)` $= 8$, so the extra monomial is X^8T^9.
- `UU` evaluates at $T = 1$, giving $x^{17} + 1 + x^8$. This will be part of the affine equation $R(x,y,z) = 0$.

4 Template of the invariant 2-jet differential

```
Jetyx := expand(
+ (1/Rz^(m-0*2)) * add(AA||j * x1^(m-0*3-j) * (y1)^j * Delta_yx^0, j=0..m-0*3)
+ (1/Rz^(m-1*2)) * add(BB||j * x1^(m-1*3-j) * (y1)^j * Delta_yx^1, j=0..m-1*3)
+ (1/Rz^(m-2*2)) * add(CC||j * x1^(m-2*3-j) * (y1)^j * Delta_yx^2, j=0..m-2*3)
+ (1/Rz^(m-3*2)) * add(DD||j * x1^(m-3*3-j) * (y1)^j * Delta_yx^3, j=0..m-3*3)
):
```

This constructs the general local form of an invariant 2-jet differential on the chart $\{R_z \neq 0\}$ (Proposition 5.1):

$$J_{yx} = \sum_{k=0}^{\lfloor m/3 \rfloor} \frac{1}{R_z^{m-2k}} \sum_{j=0}^{m-3k} F_{j,k}(x,y,z)\,(y')^j\,(x')^{m-3k-j}\,\big(y'x'' - y''x'\big)^k.$$

- `x1`, `x2` denote x', x''; `y1`, `y2` denote y', y''.
- `Delta_yx` is the Wronskian $y'x'' - y''x'$.
- `AA||j`, `BB||j`, `CC||j`, `DD||j` are undetermined polynomials.

5 Degree bounds for the coefficient polynomials

```
dA := (m-0*2)*d - m + twist:
dB := (m-1*2)*d - m + twist:
dC := (m-2*2)*d - m + twist:
dD := (m-3*2)*d - m + twist:
```

Each coefficient $F_{j,k}$ (here labelled AA_j, BB_j, CC_j, DD_j) will be a polynomial with total degree in x, y, z bounded by the values `dA`, `dB`, etc. These bounds come from the homogeneous degree constraints described in the paper.

6 Wronskian and change of chart formulas

```
Delta_yx := y1*x2 - y2*x1:
Soly1 := - x1*Rx/Ry - z1*Rz/Ry:
Soly2 := -x1^2*Rxx/Ry+2*x1^2*Rxy*Rx/Ry^2+2*x1*Rxy*z1*Rz/Ry^2
         -2*x1*z1*Rxz/Ry-Ryy*x1^2*Rx^2/Ry^3-2*Ryy*x1*Rx*z1*Rz/Ry^3
         -Ryy*z1^2*Rz^2/Ry^3+2*z1*Ryz*x1*Rx/Ry^2+2*z1^2*Ryz*Rz/Ry^2
         -z1^2*Rzz/Ry-x2*Rx/Ry-z2*Rz/Ry:
```

- `Delta_yx` is the Wronskian $y'x'' - y''x'$.
- `Soly1` gives y' from the relation $x'R_x + y'R_y + z'R_z = 0$.
- `Soly2` gives y'' after substituting y' into the second derivative relation.

7 Transition to the other affine chart $(R_y \neq 0)$

```
Jetzx := sort(mtaylor(
  expand(subs(x2=(Deltazx+x1*z2)/z1,
  expand(subs({y1=Soly1,y2=Soly2}, Jetyx)))),
  [x1,z1,Deltazx], 100), [x1,z1,Deltazx]):
```

We substitute `y1`, `y2` and replace `x2` using the relation $x_2 = (\Delta_{zx} + x_1 z_2)/z_1$ obtained from the Wronskian $\Delta_{zx} = z'x'' - z''x'$. The result is expanded as a Taylor series in x', z', Δ_{zx} up to order 100. This yields the expression of the jet differential on the chart $\{R_y \neq 0\}$.

8 Extract coefficients after clearing denominators

```
printlevel := 2:
for k from 0 to m/3 do
for j from 0 to m-3*k do
  EEEq[m-j-3*k,j,k] := factor(
    Rz^(m-j-3*k) * Ry^(m-2*k)
    * coeftayl(Jetzx, [x1,z1,Deltazx]=[0,0,0], [m-j-3*k,j,k])):
od: od:
```

For each triple (i, j, k) with $i + j + 3k = m$, we extract the coefficient of $(x')^i(z')^j(\Delta_{zx})^k$ and multiply by $R_z^i R_y^{m-2k}$ to clear allowed denominators. The result `EEEq[i,j,k]` is a polynomial that must satisfy divisibility conditions by R_z in the quotient ring $\mathfrak{R} = \mathbb{C}[x, y, z]/(R)$.

9 Define the explicit surface and its partial derivatives

```
RR := z^(d+1) + y^(d+1) + UU;
Rx := diff(RR,x): Ry := diff(RR,y): Rz := diff(RR,z):
Rxx := diff(RR,x,x): Rxy := diff(RR,x,y): Rxz := diff(RR,x,z):
Ryy := diff(RR,y,y): Ryz := diff(RR,y,z): Rzz := diff(RR,z,z):
```

The affine equation is

$$R(x, y, z) = z^{17} + y^{17} + x^{17} + x^8 + 1.$$

All first and second order partial derivatives are computed symbolically.

10 Truncation in z and preparation for y-reduction

```
printlevel := 2:
for k from 0 to m/3 do
for j from 0 to m-3*k do
  EEEq[m-j-3*k,j,k] := mtaylor(EEEq[m-j-3*k,j,k], z, (m-j-3*k)*d):
od: od:
```

Because the divisibility condition requires that each `EEEq` be divisible by $z^{(d-1)i} = z^{16i}$ (since $R_z = 17z^{16}$), we truncate the Taylor expansion in z at order $i \cdot d$ (where $i = m - j - 3k$). This truncation will later be used to collect vanishing coefficients.

11 Reduction of high y-powers using the surface equation

```
for j1 from 0 to 1+m*d+d do
  ny[j1] := (-z^(d+1)-UU)^(floor(j1/(d+1))) * y^(j1-(d+1)*floor(j1/(d+1))):
od:
```

In the quotient ring $\mathfrak{R}$, the relation $y^{d+1} = -z^{d+1} - x^{d+1} - x^8 - 1$ allows us to replace any monomial y^{j_1} (with $j_1 \geq d+1$) by a polynomial of y-degree $< d+1$. The array `ny[j1]` gives the reduced form of y^{j_1}: for $j_1 < d+1$ it is simply y^{j_1}; otherwise we factor out $(y^{d+1})^q$ and substitute the right-hand side. This implements Lemma 5.2 of the paper.

12 Expand the unknown polynomials as functions of y only (first stage)

```
for l from 0 to m-0*3 do  AA||l := add(A||l[j]*y^j, j=0..min(d,dA)):  od:
for l from 0 to m-1*3 do  BB||l := add(B||l[j]*y^j, j=0..min(d,dB)):  od:
for l from 0 to m-2*3 do  CC||l := add(C||l[j]*y^j, j=0..min(d,dC)):  od:
for l from 0 to m-3*3 do  DD||l := add(D||l[j]*y^j, j=0..min(d,dD)):  od:
```

At this stage the coefficients are written as polynomials in y alone (their coefficients are still undetermined functions of x and z). The degree in y is bounded by $\min(d, \text{total degree bound})$.

13 Reduce the equations in y-degree and extract coefficients

```
printlevel := 2:
for k from 0 to m/3 do
for j from 0 to m-3*k do
  degree(EEEq[m-j-3*k,j,k], y):
  EEEq[m-j-3*k,j,k] := mtaylor(EEEq[m-j-3*k,j,k], y, 1+m*d+d):
  dy[m-j-3*k,j,k] := degree(EEEq[m-j-3*k,j,k], y):
od: od:
```

We expand each `EEEq` in y up to a sufficiently high order and record the resulting y-degree. Then for each power of y we extract its coefficient and substitute the reduction `ny[j1]`:

```
printlevel := 3:
for k from 0 to m/3 do
for j from 0 to m-3*k do
  for j1 from 0 to dy[m-j-3*k,j,k] do
    Cy[m-j-3*k,j,k,j1] := coeff(EEEq[m-j-3*k,j,k], y, j1):
  od:
od: od:
```

```
printlevel := 2:
for k from 0 to m/3 do
for j from 0 to m-3*k do
  EEq[m-j-3*k,j,k] := add(Cy[m-j-3*k,j,k,j1]*ny[j1], j1=0..dy[m-j-3*k,j,k]):
  degree(EEq[m-j-3*k,j,k], y):
od: od:
```

The new `EEq[m-j-3*k,j,k]` is now a polynomial with y-degree strictly less than $d+1$, representing the original expression in the normal form required by Lemma 5.2.

14 Introduce z-dependence into the unknown coefficients

```
printlevel := 2:
for l from 0 to m-0*3      do
for j from 0 to min(d,dA) do
  A||l[j] := add(A||l[j,k]*z^k, k=0..dA-j):
od: od:
...
```

Now the coefficients that were earlier just polynomials in y are expanded as polynomials in z (with undetermined coefficients depending on x). Similar blocks for BB, CC, DD.

15 Truncate equations in z after y-reduction

```
printlevel := 2:
for k from 0 to m/3 do
for j from 0 to m-3*k do
  EEq[m-j-3*k,j,k] := mtaylor(EEq[m-j-3*k,j,k], z, (m-j-3*k)*d):
  degree(EEq[m-j-3*k,j,k], z):
od: od:
```

The same truncation in z is applied again because the y-reduction may have introduced higher powers of z.

16 Introduce x-dependence into the unknown coefficients

```
printlevel := 3:
for l from 0 to m-0*3      do
for j from 0 to min(d,dA) do
for k from 0 to dA-j       do
  A||l[j,k] := add(A||l[i,j,k]*x^i, i=0..dA-j-k):
od: od: od:
...
```

Now the undetermined coefficients are expanded as polynomials in x. At this point every unknown is a scalar constant $A_{l,i,j,k}$, etc.

17 Display the number of terms in each coefficient

```
for l from 0 to m-0*3 do nops(expand(AA||l)) od;
for l from 0 to m-1*3 do nops(expand(BB||l)) od;
for l from 0 to m-2*3 do nops(expand(CC||l)) od;
for l from 0 to m-3*3 do nops(expand(DD||l)) od;
```

These numbers are still large (e.g. 10914 at the start) but will decrease as we solve the divisibility systems.

18 Define the final truncated equations Eq

```
printlevel := 2:
for k from 0 to m/3 do
for j from 0 to m-3*k do
  Eq[m-j-3*k,j,k] := mtaylor(EEq[m-j-3*k,j,k], z, (m-j-3*k)*d):
  degree(Eq[m-j-3*k,j,k], z):
od: od:
```

`Eq[i,j,k]` is the polynomial that must be divisible by $z^{(d-1)i}$; all its coefficients of $z^0, \ldots, z^{i(d-1)-1}$ must vanish.

19 Iterative solving of divisibility by $z^{\ell d}$

The core computation is an induction on $\ell = 1, 2, \ldots, 11$. At step ℓ, we require that all $\mathtt{Eq}[i,j,k]$ with $i = \ell$ be divisible by $z^{\ell d}$. We collect the coefficients of monomials $x^p y^q z^r$ with $r < \ell d$ and set them to zero. The resulting linear system is solved and the solutions are assigned globally.

First iteration ($\ell = 1$): divisibility by $z^{1 \cdot d}$

```
Systemez1d := {}:
for k from 0 to (m-1)/3 do
  Coeffs_z1d[1,m-1-3*k,k] := {coeffs(expand(mtaylor(Eq[1,m-1-3*k,k], z, 1*d)), [x,y,z])}:
  Systemez1d := Systemez1d union Coeffs_z1d[1,m-1-3*k,k]:
od:
Systeme_z1d := Systemez1d;
Variables_z1d := indets(Systeme_z1d):
Solutions_z1d := solve(Systeme_z1d, Variables_z1d):
assign(Solutions_z1d):
```

After solving, the number of terms in each coefficient is printed; they remain large (still 10914 etc.). The same pattern is repeated for $\ell = 2, 3, \ldots, 10, 11$. At each step the systems get larger and the number of free parameters decreases. By the end of the 11th iteration, all `nops` become 1, meaning every coefficient polynomial has been forced to zero.

20 Holomorphicity at infinity

After the affine divisibility conditions are satisfied, we must check that the differential extends holomorphically across the divisor at infinity $\{T = 0\}$.

Homogenisation of the coefficients

```
for l from 0 to m-0*3 do
  HA||l := add(add(add(A||l[i,j,k]*X^i*Z^k*Y^j*T^(dA-i-j-k), i=0..dA-j-k), k=0..dA-j),
      j=0..min(d,dA)):
od:
...
```

Each coefficient is homogenised to a homogeneous polynomial in T, X, Y, Z of degree `dA` etc. The results are stored in `HA||l`, `HB||l`, etc.

Form the combinations appearing in the transition formula

```
for h from 0 to m-3*0 do
  HAEEq[h] := add(HA||l*binomial(l,h)*X^(m-3*0-l)*Y^(l-h), l=h..m-3*0):
  dYA[h]   := degree(HAEEq[h], Y):
od:
...
```

These sums implement the relation $\sum_{j=\ell}^{m-3k} \widehat{K}_{j,k}\binom{j}{\ell} X^{m-3k-j} Y^{j-\ell}$ that must be divisible by $T^{m-3k-\ell}$.

Reduce high Y-powers using the homogeneous equation

```
for j1 from 0 to 1+m*d+d do
  nY[j1] := (-Z^(d+1)-HUU)^(floor(j1/(d+1))) * Y^(j1-(d+1)*floor(j1/(d+1))):
od:
```

Analogous to the affine reduction, this replaces Y^{j_1} by lower powers using $Y^{d+1} = -Z^{d+1} - X^{d+1} - X^8T^9 - T^{d+1}$.

Substitute the reduction and truncate in T

```
for h from 0 to m-3*0 do
  HAEq[h] := expand(mtaylor(add(HAEEq[h]*nY[j1], j1=0..dYA[h]), T, m-3*0-h)):
od:
...
```

The reduced polynomial is expanded and then truncated at order $m - 3k - \ell$ in T.

Collect the coefficients and solve

```
HSysteme := {seq(coeffs(HAEq[h],[T,X,Y,Z]),h=0..m-3*0)}
        union {seq(coeffs(HBEq[h],...),...} minus {0}:
HSolutions := solve(HSysteme):
assign(HSolutions):
```

All coefficients of the truncated expressions are set to zero. Solving this final linear system removes any remaining degrees of freedom.

21 Final verification

```
Jetyx := factor(
+ (1/Rz^(m-0*2)) * x^(-twist) * add(AA||j * x1^(m-0*3-j) * (y1)^j * Delta_yx^0, j=0..m-0*3)
+ (1/Rz^(m-1*2)) * x^(-twist) * add(BB||j * x1^(m-1*3-j) * (y1)^j * Delta_yx^1, j=0..m-1*3)
... ):
```

The jet differential is rebuilt, this time including the factor $x^{-\text{twist}} = x^6$ to account for the twisting.

```
Vars := indets(Jetyx) minus {x, x1, x2, y, y1, y2, z};
NbVars := nops(Vars);
```

The output `NbVars = 0` confirms that every undetermined constant has been forced to zero, hence the only global section is the trivial one.

22 Conclusion

The computation verifies

$$H^0\big(X,\, E_{2,6}T_X^* \otimes \mathcal{O}_X(-6)\big) = 0$$

for the specific Fermat-type surface of degree 17. By semicontinuity, the same vanishing holds for a generic smooth surface of degree 17. Adjusting the parameters `m, twist, d` at the top of the code permits verification of all required cases of Lemma 1.1 for degree 17.